

Paper 091-2007

Graphing Longitudinal Data: Deaf Children in the “Children of the National Longitudinal Survey of Youth”

Paul Martino, RIT/NTID Department of Research & Teacher Education, Rochester, NY

Sara Schley, RIT/NTID Office of the Vice President & Dean, Rochester, NY

Richard Dirmyer, RIT/NTID Department of Research & Teacher Education, Rochester, NY

ABSTRACT

Historically, deaf education in the United States has achieved poor results. An oft-quoted statistic is that on average, deaf students graduating from high school (at age 18-21) perform at the level of hearing 8-10 year olds in terms of reading and writing skills (Allen, 1994; Traxler, 2000). Performing secondary analyses on the Children of the National Longitudinal Survey of Youth (NLSY) focusing on developmental and educational outcomes allows one to predict growth patterns in each for deaf and hearing children. Children were flagged as deaf by careful deliberation of multiple questions to indicate hearing status. A sibling of each deaf child was picked as a means of comparison, to control for home environment. The use of frequencies, means, and growth models disproves many past assumptions and raises some interesting questions.

INTRODUCTION

The ability to study change has fascinated researchers for many years. Change is a part of everyday life due to people age, learn new skill sets, and experience life-altering events. According to Singer and Willett “measuring and charting changes like these – both naturalistic and experimentally – we uncover the temporal nature of development. (Singer 3)”

There exist many ways to measure change for data series. These include individual growth models, random coefficient models, multilevel models, mixed models, and hierarchal linear models. However, while the ability to study change is possible, the ability to do it well requires longitudinal data. Longitudinal data addresses important questions on how people change over time along with the individual differences in change. For a longitudinal data set to be valid for a study of change, the following three characteristics are important:

- Three or more waves of data
- An outcome whose values change systematically over time
- A sensible metric for clocking time (Singer 9)

Each characteristic allows you to answer the necessary questions posed above. You can study change for differences in time and differences among individuals. Through creating a “person-period” format, where each person has multiple records for each wave you can run the required tests to study longitudinal change and differences.

NATIONAL LONGITUDINAL SURVEY

It is important to note that the data set used consists of a secondary sample that came from the National Longitudinal Survey (NLS). THE NLS was given to four different samples. These samples, or cohorts, represent a population at a given time. These cohorts include two nationally representative samples of women (ages 14-24 and 30-44) and of men (ages 14-24 and 45-59). Policy makers first studied these age ranges between 1966 and 1968 because they were each at critical time points in making important labor market decisions. The selection of the 14-24 year old cohort investigated the role of family factors, school performance, and early work experience on the employment trajectories of young adults.

In 1979, a new and fifth cohort was selected of 14-21 year old young adults, called the National Longitudinal Survey of Youth 1979 (NLSY79). The survey asked these young adults multiple questions to evaluate job training and employment opportunities for themselves. The sample included 12,686 men and women and was collected from the period 1979 to 1994 annually and 1996 to 2002 biannually.

In 1986, the NLS added an addition to the fifth cohort that included the children of the mothers from the NLSY79 survey. This adds a second generation to the NLSY survey. The survey went from 1986 to today biannually. We focus on eight waves of data from 1986 to 2002. In 1986, 5,225 children responded to the surveys and through active participation and insertion of new children constantly, 6,417 responded in 2000. Looking at such a data set allows you to see the impact the lives of mother and child have upon one another.

Our goal in this presentation is to illustrate the power of using SAS/SAS® Graph tables and graphs to accurately analyze complex longitudinal data. In particular, we will illustrate visual displays of individual growth curves and summary models. The individual curves inform and extend the analysis beyond more typical presentation of grouped summary information.

CHILDREN OF THE NLSY79

It is important to understand why children of the NLSY79 data were picked. Typically, longitudinal studies of deaf children focus on children in non-mainstream settings in order to find a sizeable sample. By conducting secondary analyses of an ongoing large scale educational dataset like the NLSY, we can study the children of a nationally representative sample of mothers. Additionally, the NLSY allows for some economies of scale in terms of both sample size as well as the length of the longitudinal follow-up time. Finally, since all children of women in the cohort were added to the sample, we are able to compare both deaf and hearing children from the same family – siblings

DEAF OR NOT DEAF? IDENTIFYING THE HARD OF HEARING SAMPLE

Before any deeper analysis can begin, you need to extract the sample of deaf children from the main sample. One could go two directions. The first direction is from point of view of the mother and the second being from the point of view of the interviewer. Each point of view corresponds to different questions. From 1986 – 2002, the survey asked questions from the “Health” section for each year to gauge if a child was deaf or not. Those questions were “Health: Does child have serious hearing difficulty or deafness” for years 1986 – 1994 and “Health: What limiting condition does child have” for years 1994 – 2002. These questions coincide with the mother’s point of view.

The question had multiple responses to account for a variety of disabilities. If the child’s limiting condition was deaf, the corresponding response was 7. If the child was not deaf, the response was 0. If missing or unknown, the response was -7 or -1 respectively. Looking at each year and the patterns amongst answers for each child, 21 deaf children were found using this method. Afterwards, additional questions were analyzed. These questions contained flags to look at child’s hearing characteristic in different ways such as use of special equipment, participation in ESL courses, and cancellation of tests due to hearing problems, among others. The reason for this step was to make sure the children were deaf along with flagging children missed as deaf in the initial phase. After this step was taken there were 36 children flagged as deaf.

Interviewers were also asked to evaluate whether they noticed any problems with the child’s hearing during the data collection process. Looking at the questions “Testing Conditions: Any problems with child’s hearing?” for years 1988 – 2002 one can extract a different set of children from the point of view from the interviewer. The same procedure as before was performed and 33 additional children were labeled identified with hearing problems. Therefore, 65 children total are flagged deaf or identified with hearing problems. For simplicity, we refer to this group as the “deaf” sample throughout this paper – although we also kept a variable in the dataset tagging the source of this identification (mother or interviewer) in order to run sub-group comparisons.

FINALIZING THE SAMPLE

To finish the sample, siblings of the deaf children need to be extracted as well. The main point of this analysis is a comparison between hearing and deaf children. Therefore, siblings need to be as close as possible to the deaf children for comparison. The understanding is that a hearing child of age five is comparable to a deaf child of age five, regardless of year born. Since there are multiple waves of data collection in the NLSY – we were able to control for age carefully by choosing the sibling with the smallest age difference from the deaf child – whether or not the data was collected during the same wave or time point. For example – if a deaf child was 4 years old in 1986, and they had a 2 year old sibling at the time – that sibling was 4 years old in the 1988 wave. Thus, the deaf and hearing siblings’ outcome measures were compared during those two waves – controlling for age, rather than calendar year.

Children NLSY79 ID codes allow quick and easy classification of child’s mother and sibling number. Each ID consists of the mother’s ID used in the NLSY79 survey along with the child’s id number concatenated to it. This child

id is a double-digit number with 01 as the first child and 10 as the tenth.

Going back to the NLSY79 Child/Young Adult Survey and using the questions “Child Background: Age of Child (in months) at Child Assessment Date, Mother Supplement” and “Child Background: Age of Child (in years) at Child Assessment Date, Mother Supplement” from 1986 – 2002. Years and months are both picked since while years is an easy comparison there may come a time when two siblings are equally comparable and we need to use months to see more closely which child is more comparable in age. One sibling is picked for each child unless twins. This gives a total sample size of 126 children. There are 65 deaf children and 61 hearing siblings (four of the deaf children were singletons).

The NLSY Child/Young Adult Survey also contains other descriptive questions. These include age, gender, mother's age, mother's age at birth, and family size. To adequately show change in the children, a number of tests were given in the survey that are used to showcase the educational and social development of deaf and hearing children. After all the data is obtained from the survey, the data was formatted into a “person-period” format that contains an (1) subject identifier, (2) time indicator, (3) outcome variables and (4) predictor variables. Table 1 is an example of this. By setting the data set in this format, it allows for a varying amount of waves for each child and time-varying predictors. This is necessary for longitudinal analysis where each subject does not have the same number of waves. Since you assume change will occur, the predictor must be able to vary by time.

| ID | TIME | DEAF | SEX | PIAT MATH |
|--------|------|------|-----|-----------|
| 102504 | 2 | 0 | 0 | 44 |
| 102504 | 4 | 0 | 0 | 16 |
| 102504 | 6 | 0 | 0 | . |
| 113301 | 8 | 0 | 0 | 43 |
| 113302 | 0 | 1 | 0 | 48 |
| 121003 | 2 | 1 | 1 | . |
| 121003 | 4 | 1 | 1 | 4 |
| 121003 | 6 | 1 | 1 | 12 |
| 121003 | 8 | 1 | 1 | 9 |

Table 1: “Period-Period” Data Set

DESCRIPTIVE ANALYSIS OF THE SAMPLE

The paper does not solely focus on deaf and hearing students, but also on gender, race, early education, family size, etc. that can be used in the construction of a model that can predict educational scores. A descriptive analysis allows you to analyze the data set and point out any discrepancies instantly before the construction of the full model. These averages allow you to gain a basic overview of the data set, along with a basic comparison of deaf and hearing children. However, means and frequencies do not give enough information and further analysis is required.

EMPIRICAL GROWTH PLOTS

Once the differences between deaf and hearing children are established, quantifying this difference is the next step. Empirical growth plots are an effective way to determine how a person changes. Using these plots, you can measure growth of children for a range of educational test scores. To illustrate, using the PIAT Math test as the outcome variable and age as the time indicator, you can create empirical growth plots in Figure 1 using this SAS® code.

```
goptions reset= all noborder nocell rotate = portrait htext = 1 ftext = "arial" hby
= 0 vsize = 8 hsize = 8;
axis1 label = ('Current Age (years) - 5') order = (0 to 10 by 2) minor = none;
axis2 label = none order = (0 to 90 by 15) minor = none;
symbol color = black value = dot height = 2.5;
```

The GOPTIONS command is used consistently for all the graphs throughout the table. The options controls the size of the text, size of the graphs, labeling of the axis, and the font used. These options are assumed to be in use for each presented throughout with exceptions done in SAS® code.

```

proc greplay igout = work.gseg nofs;
  delete _all_; run;
proc gplot data= newage uniform;
  where id in ( 24001, 24002, 304602, 304605, 396201, 396204, 413202, 413203,
  527103, 527102, 568702, 568703, 642902, 642903, 674201, 674202);
  by id;
  note justify=right 'ID #byval(ID)';
  plot piat_math*age_years / nolegend haxis=axis1 vaxis=axis2 noframe; run;

```

This step creates a graph for each id in the data file regardless of number.

```

proc greplay igout=work.gseg tc=sashelp.templt template=h4 nofs;
  treplay 1:1 2:2 3:3 4:4;
  treplay 1:5 2:6 3:7 4:8;
  treplay 1:9 2:10 3:11 4:12;
  treplay 1:13 2:14 3:15 4:16; run;

```

You then create a template. Each TREPLAY places four graphs on a line for four templates in total. Four is an arbitrary number. You can increase or decrease the number of graphs per line.

```

proc greplay igout=work.gseg tc=sashelp.templt template=v2 nofs nobyline;
  treplay 1:template 2:templat1;
  treplay 1: templat2 2: templat3; run;
proc greplay igout=work.gseg tc=sashelp.templt template=v2 nofs nobyline;
  treplay 1:templat4 2:templat5; run;

```

Finally, after creating a template that combines two sets of four in a set of eight, we combine those two sets into sixteen graphs seen below.




Figure 1: Empirical Growth Plots for Sixteen Children

Figure 1 shows obvious change in PIAT MATH scores as children get older. The PIAT MATH test is given to children starting at age five to age fourteen, which is the last year children take the survey. As children increase in age, there is a definite positive increase in test scores. Looking at Figure 1 you can see who has the highest test scores, who does not improve, who appears to change the quickest, etc. Some children appear to have sharp linear trends and quickly improve over time. Others have improved and then appear to level out. However, performing an empirical analysis on every child to gain insight is not only time-consuming but ineffective. There is no discernible way to measure this change nor is it effective in comparing differences among individuals and siblings. Further analysis is required.

EXPLANATORY ANALYSIS THROUGH REGRESSION CURVES

The next step in the process involves looking at the change among individuals and siblings. There are two steps for this process. The first is connecting a smooth line for each point. This line allows comparison of the trend that exists. This line could be linear or curvilinear for example. It also allows you to notice if certain increases are sharper than others. The second step involves using ordinary least squares regression. This entails drawing a line between the points that is an accurate prediction model. However, due to the size of our sample, there would be 121 individual regression plots. This method could be used to compare changes among all children but it would be quite difficult to accurately compare differences among deaf and hearing children. By layering all the individual regression lines on one graph, the process is simplified. Note that here; we are treating each individual child and their growth over time as an individual regression line.

```
proc gplot data = newage;
  plot piat_math * time = id / nolegend haxis = axis1 vaxis = axis2;
  symbol v= none repeat = 126 i = sm50s color = grey width = 1;
  label time = 'Current Age - 5';run;
```




Figure 2: Individual PIAT Math Test Results

Combining all the empirical growth plots onto one graph as shown in Figure 2 allows you to see trends and differences otherwise not notable. For example, there is a definite linear relationship between math test results and age. Between the ages of 5 and 9 most of the test results are grouped together. Children appear to perform equally at these ages. It is not until after 9 that children start to separate. There are two groups of children that appear to perform either better or worse than the average group of children. Obtaining information from the graph is difficult due to the number of lines. Using Figure 2 and fitting an average trend line for all the children, you can see how each child compares to all of the children on average as shown by Figure 3.

```
proc gplot data = newage;
  plot piat_math * time = id / haxis = axis1 vaxis = axis2 noframe nolegend;
  plot2 piat_math * time / vaxis = axis2 noframe;
  symbol v= none repeat = 121 i = sm50s color = gray height = 2;
  symbol2 v= none color = black i = sm50s width = 4;
```




Figure 3: Individual PIAT Math Test Results with a Fitted Average Trend Line

The average trend line shows that the change in math scores has a curvilinear slope that increases slowly, about 15 points every two years. It then starts to level off around age 9. This appears to be the age where children start breaking away from the trend. Now you know how each child compares to each other and to the average child. You can compare deaf and hearing children in a similar process. By grouping two average trend lines, one for deaf and one hearing, you can gain insight on how the average deaf and hearing child compare shown by Figure 4.

```
proc format;
  value deafgrp      1 = 'Deaf Children' 0 = 'Hearing Children'; run;
proc gplot data = newage;
  plot piat_math * time = id / noframe haxis = axis vaxis = axis2 nolegend;
  plot2 piat_math * time = deaf / noframe vaxis = axis2;
  symbol1 v = none repeat = 121 i = sm50s color = darkgray height = 2;
  symbol2 v = none i = sm50s color = purple width = 4 line = 1;
  symbol3 v = none i = sm50s color = orange width = 4 line = 1;
  format deaf deafgrp.; run;
```




Figure 4: Individual PIAT Math Test Results with Fitted Trend Lines for Deaf and Hearing Children

Figure 4 shows that hearing children score higher compared to deaf children. Hearing children increase at a faster curvilinear rate. The differences become predominate at age 7. Deaf children while scoring lower still show a slight

linear trend. At age 9, the scores tend to level out. There appears to be some outliers in the deaf sample as in the even years starting at 10, there are slight increases in the curve. NLS offers the survey every two years, which causes this pattern to occur. Most likely one or two deaf children are performing at a much higher level than the average deaf student is. Looking at Figure 3 and Figure 4, you can see it overall trend line falls in between the trend lines for deaf and hearing children as expected.

You can perform a deeper exploratory description of the data set by using interactive variables. For example if you would like to see the difference between deaf males and deaf females you can create a similar looking graph as Figure 4. However, there will be four lines for each interaction between deaf and gender.

```
proc format;
  value deafsexgrp      1 = 'Hearing Male' 2 = 'Deaf Male'
                      3 = 'Hearing Female' 4 = 'Deaf Female'; run;
proc gplot data = newage;
  plot piat_math * time = id / noframe haxis = axis1 vaxis = axis2 nolegend;
  plot2 piat_math * time = deafsex / noframe vaxis = axis2;
  symbol v = none repeat = 121 i = sm50s color = darkgray height = 2;
  symbol2 v = none i = sm50s color = purple width = 4 line = 1;
  symbol3 v = none i = sm50s color = orange width = 4 line = 1;
  symbol4 v = none i = sm50s color = purple width = 4 line = 2;
  symbol5 v = none i = sm50s color = orange width = 4 line = 2;
  format deafsex deafsexgrp.; run;
```

Individual PIAT Math Test Results with Fitted Average Trend Lines for Children including Hearing Condition and Gender




Figure 5: Individual PIAT Math Test Results with Fitted Trend Lines for Children Including Hearing Condition and Gender

You can see in Figure 5, on average, hearing males perform the best and the worst. Hearing males have the highest scores. Deaf males have lower than hearing males and females. Females, either deaf or hearing, performed similar over the period.

LONGITUDINAL DATA ANALYSIS

To understand population behavior, you can create statistical models to represent such behavior. These models consist of trends, intercepts, slopes, variances, etc. that create a general overview of how the population behaves. Since the data set is longitudinal, a multilevel statistical model needs to be created that can answer two types of questions: level-1 questions that answer “within-person” change and level-2 questions that answer “between-person” differences in change (Singer 47). The level -1 submodel describes how individuals change over time and the level-2 submodel describes how these changes vary across individuals.

You can see the level-1 submodel by using the empirical growth in Figure 1. You can do this by running ordinary least square trajectories for each id.

```
proc gplot data=newage uniform;
  where id in ( 24001, 24002, 304602, 304605, 396201, 396204, 413202, 413203,
    527103, 527102, 568702, 568703, 642902, 642903, 674201, 674202);
  by id;
  note justify=right 'ID #byval(ID)';
  plot piat_math * time / nolegend haxis=axis1 vaxis=axis2 noframe;
  symbol color = black interpol = reg value = dot height = 2; run;
```




Figure 6: Empirical Plots Fitted with OLS Trajectories

Figure 6 uses the same code as Figure 1 but with an additional SYMBOL command. You can draw OLS trajectories to estimate the population behavior. A portion of the output is included to display the change. These OLS trajectories give a basic statistical model: age predicts PIAT MATH score. The linear nature of the plots above shows this; PIAT MATH scores do change over time. However, the level-2 questions still need to be answered. How do the changes over time vary amongst individuals?

The level-2 submodel has four distinct features that separate it from the level-1 submodel. Those features include:

- Outcomes must be individual growth parameters
- Level-2 submodel must be written in separate parts, one for each level-1 growth parameter
- Each part must specify a relationship between an individual growth parameter and the predictor,
- Each model must allow for individuals who share common predictor values to vary in their individual change trajectories (Singer 58).

Computer software currently allows data to fit the multilevel model for change. SAS® offers two routines, PROC MIXED and PROC NL MIXED to fit multilevel models for change. Due to size and topic, detailed information for the construction and specifications of models is not included. However, here is an example of a model that can be run using PROC MIXED.

```
proc mixed data=newage method=ml noclprint noinfo covtest;
  class id;
  model piat_math = deaf sex TIME deaf*TIME sex*TIME deaf*sex /solution notest;
  random intercept TIME/type=sp(lin) (time) subject=id; run;
```

The code above is a simple multilevel model only testing the significance of hearing condition, gender, and their interactions over time on PIAT MATH test scores. The CLASS statement keeps each id as a separate entity. The RANDOM statement declares the intercept random with a linear spatial covariance structure along with ID declared as random. You can change the covariance structure dependent on the type of data set. You can use PROC MIXED for unbalanced data as well. Unbalanced data can cause problems with convergence if there is not enough information or too many people have a few numbers of waves. However, in this case, there is a good amount of information and you can use PROC MIXED without worry. Now that you have the multilevel of change you can use that model to fit values to it and create an idea how children will score. Using a DATA step, a data file is created that contains DEAF, SEX, TIME, and GROUP ('Hearing Male', 'Deaf Female', etc.). TIME is either 0 or 10. These values are plugged into the model in the code below and are used to estimate PIAT MATH scores.

```

data figsa;
  set fig;
  modell = 9.3017 + 6.5305*deaf + 5.4421*sex + 4.7118*time -1.0327*deaf*sex
  13.8337*deaf*sex; run;

goptions reset = all hsize = 8 vsize = 8 hby = 0 ftext = "arial"
gsfname = outgraph dev = gif570 rotate = portrait colors=(black);
symbol1 i = join l = 2 w = 2 color = orange;
symbol2 i = join l = 1 w=2 color = orange;
symbol3 i = join l= 2 w= 2 color = purple;
symbol4 i = join l= 1 w= 2 color = purple;
axis1 order = (0 to 9 by 1.5) minor = none label=('TIME');
axis2 order = (0 to 90 by 15) minor = none label=none;
proc gplot data = figsa;
  plot modell*time = group /noframe haxis = axis1 vaxis = axis2; run;

```




Figure 7: Fitted Multilevel Model Plots Dependent on Hearing Condition and Gender

The above graph is estimates for each child using the multilevel model of change. On average, a corresponding child will score the same as the model dictates.

CONCLUSION

By using SAS® and its graphing capabilities exploratory analysis of longitudinal data is for more capable then other methods. Comparing means between groups is too weak of a comparison and does not allow you to see trends or track change. OLS trajectories or level-1 models, do not track differences among individuals and only shows the change over time. SAS® can help visualize the multilevel model by fitting group means overlaid on individual plots. This visualization makes longitudinal analysis easier to grasp.

REFERENCES

Singer, Judith D., and John B. Willett. 2003. *Applied Longitudinal Data Analysis*. New York, New York: Oxford University Press, Inc.

Allen, T.E. 1994. "Who are the deaf and hard-of-hearing students leaving high school and entering postsecondary education?" *A Comprehensive Evaluation of the Postsecondary Educational Opportunities for Students who are Deaf or Hard of Hearing*. U.S. Office of Special Education and Rehabilitative Services. <<http://gri.gallaudet.edu/AnnualSurvey/whodeaf.html>> (March 1, 2007)

Traxler, Carol Bloomquist. 2000 Fall. "The Stanford Achievement Test, 9th Edition: National Norming and Performance

Standards for Deaf and Hard-of-Hearing Students". *Journal of Deaf Studies and Deaf Education*. 5(4): 337-348.

Bureau of Labor Statistics. "National Longitudinal Surveys". <<http://www.bls.gov/nls>>

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Paul Martino
Enterprise
152 Barrington Street Apartment 209
Rochester, NY, 14607
585-224-5933
E-mail: pmartino@gmail.com

Sara Schley
RIT/NTID Office of the Vice President & Dean
52 Lomb Memorial Drive
Rochester, NY 14623
Work Phone: 585-475-7981
E-mail: sara.schley@rit.edu

Rich Dirmyer
RIT/NTID Department of Research and Teacher Education
96 Lomb Memorial Drive
Rochester, NY 14623
585 -259-2852
Email: rcd0275@rit.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Other brand and product names are trademarks of their respective companies.